*Computers & Geosciences* Vol. 7, pp. 115–122
Pergamon Press Ltd., 1981. Printed in Great Britain

0098-3004/81/0301-0115/$02.00/0

# XRDPLT: A FORTRAN IV PROGRAM FOR THE GRAPHICAL REPRESENTATION OF X-RAY POWDER DIFFRACTION DATA†

C. B. JONES and R. I. LAWSON
Institute of Geological Sciences, Murchison House, West Mains Road, Edinburgh, Scotland

(*Received* 12 *February* 1980)

## INTRODUCTION

The identification of crystalline substances from X-ray powder diffraction patterns on film may involve measuring the diffraction line positions and intensities, calculating the equivalent interplanar spacings $d$ and comparing these $d$-spacings with data recorded in reference tables. A rapid method of confirming the suspected presence of a certain phase is to compare its X-ray powder diffraction pattern on film with those of films of standard reference substances. In the absence of standard films for reference however, the process of matching with numerical data can be simplified by representing the standard data in an appropriately scaled plot, upon which the unknown film may be superimposed to enable a rapid comparison between the two. In addition, the use of sets of such plots of the numerical data provides a visual aid to the comparison of mineral structures within a mineral group.

## THE RELATIONSHIP BETWEEN DIFFRACTION LINES AND LATTICE SPACING

The locations of the arcs, or "lines", on an XRD film are a function of the geometry of the diffracting lattice planes of the crystal, the wavelength of the incident radiation and the diameter of the X-ray camera. The relationship between lattice spacing $d$, wavelength $\lambda$ and the angle of incident radiation to the lattice plane $\theta$, is given by the Bragg equation:

$$n\lambda = 2d \sin \theta. \qquad (1)$$

The measured distance $s$ (Fig. 1) of a film line from the axis of the incident X-rays in a cylindrical camera of radius $R$ is given by

$$s = 2\theta \cdot R. \qquad (2)$$

Therefore combining equations (1) and (2) for $n = 1$ gives

$$s = 2\,R \arcsin(\lambda/2d).$$

This equation is used throughout the computer program XRDPLT to calculate plotted positions for particular lattice spacings.




Figure 1. Relationship between diffraction line spacing on film and angle of incident X-rays.

## FUNCTION OF THE PROGRAM XRDPLT

The program XRDPLT is designed to produce graphical displays of standard XRD spacings from data sets such as those published by the Joint Committee on Powder Diffraction Standards (JCPDS, 1974). It is written in FORTRAN IV and, in the version listed here, plotted output is invoked by calls to the standard CALCOMP plôtting routines for plotter initialisation (PLOTS), pen movement (PLOT), text plotting (SYMBOL) and number plotting (NUMBER).

The plotted output from XRDPLT consists of sets of ticks marking the positions of the lines which would appear on an XRD film, and with lengths proportional to the intensities of the corresponding film lines (see Plate 1). As an aid to differentiating between lines of relatively low intensity, the program includes options for scaling the ticks in proportion to the logarithm or the square root of the intensity, as well as linearly. The advantages of nonlinear scaling can be appreciated by comparing the plots in Plate 1. All recorded lines to the right of the film exit aperture are represented and, in order to enable correct alignment in overlaying the unknown film, the equivalent lines to the left for lattice spacings greater than 2.5 Angstroms are also plotted. There is provision for representing broad bands by a tie line linking the upper and lower limiting lines of the band (Fig. 2). A scale in Angstrom units may be plotted beneath the ticks

†Published with the permission of Director, Institute of Geological Sciences.




Plate 1. XRDPLT plots of data recorded on JCPDS reference card 5-0490 for mineral quartz, for each of three intensity scaling options, and matching X-ray powder diffraction film. Film was produced with copper $K\alpha$ radiation in 114.83 mm diameter Debye–Scherrer camera. For purposes of detailed comparison film would usually be superimposed on plot.




Figure 2. XRDPLT plot of hydrated calcium silicate mineral gyrolite illustrating convention used to display "broad-band" reflections. In plot intensities are assumed to be constant throughout band. Note that overlapping bands and reflections are plotted.




Figure 3. XRDPLT plots for quartz using various wavelengths. Camera size is constant throughout.




Figure 4. XRDPLT plots of garnet group of minerals. Variations that may occur within a group of minerals are well illustrated by this method of display.

if required and, if the type of radiation and a title are included on the input data cards, this information will be read from the first card for each set of crystal data and plotted at top of the ticks. For a given camera diameter, the scale of the plot depends upon the wavelength specified, and the size of characters in the scale plotted will differ accordingly (see Fig. 3).

There is no limit to the number of data sets of substances which may be plotted, one above the other (see Fig. 4), except that imposed by the size of the plotting device.

For each substance plotted the input data required are the wavelength of the incident radiation, and the set of lattice spacings, in Angstroms, and their relative intensities, normalised to 100. In the version of the program given here, the camera diameter is assumed to be 11.483 cm. Details of the data cards required to run the program, with an example dataset, are included in the program listing in the Appendix.

*Acknowledgments*—The XRD data used in this paper is published by courtesy of JCPDS—International Center for Diffraction Data, 1601 Park Lane, Swarthmore, PA 19081, U.S.A.

## REFERENCE

JCPDS, 1974, Powder Diffraction File: Joint Committee on Powder Diffraction Standards, Swarthmore, Pennsylvania.

## APPENDIX

```
C...MAIN XRDPLT
C
C  AUTHOR:  C.B. JONES
C           INSTITUTE OF GEOLOGICAL SCIENCES
C           WEST MAINS ROAD, EDINBURGH
C
C   XRDPLT USES X-RAY DIFFRACTION POWDER PATTERN MEASUREMENTS TO
C   PRODUCE GRAPHS REPRESENTING THE POSITIONS AND INTENSITIES OF THE
C   XRD FILM LINES FOR PARTICULAR MINERALS OR MIXTURES OF MINERALS.
C
C    INPUT DATA FORMATS:
C  FIRST CARD
C  I2     COLS   1-2: THE NUMBER OF GRAPHS TO BE PLOTTED
C  A4     COLS   4-8: SET TO SCAL IF A SCALE IS REQUIRED
C  A4     COLS 11-14: INTENSITY SCALE:- SET TO LOG (LEFT JUSTIFIED)
C                     FOR LOGARITHMIC, SQRT FOR SQ. ROOT OR LEAVE
C                     BLANK FOR LINEAR
C  SECOND AND SUBSEQUENT CARDS - ONE CARD PER FILM LINE AND AN END
C  CARD AFTER EACH SET OF DATA TO BE PLOTTED
C  A4     COLS   1-4: FOUR CHARACTERS INDICATING TYPE OF RADIATION
C  F8.0   COLS  5-12: WAVELENGTH OF RADIATION IN ANGSTROMS
C  F10.0  COLS 13-22: LATTICE SPACING IN ANGSTROMS
C  F10.0  COLS 23-32: SECOND LATTICE SPACING REQUIRED FOR BROAD BANDS
C  F4.0   COLS 33-36: INTENSITY OF LINE NORMALISED TO 100
C  11A4   COLS 37-80: TITLE
C  TERMINATING CARD FOR EACH SET OF DATA TO BE GRAPHED
C  A3     COLS   1-3: THE WORD END
C
C  EXAMPLE DATASET (SHIFTED 2 COLUMNS TO RIGHT) :-
C  1 SCALE SQRT
C CUKA1.5418  22.                       1009-449,GYROLITE
C CUKA1.5418  11.0                      0809-449,GYROLITE
C CUKA1.5418  8.4         7.4           0409-449,GYROLITE
C CUKA1.5418  7.4         5.4           0209-449,GYROLITE
C CUKA1.5418  4.75                      0209-449,GYROLITE
C CUKA1.5418  4.20                      0809-449,GYROLITE
C CUKA1.5418  3.72        3.45          0209-449,GYROLITE
C CUKA1.5418  3.65                      0609-449,GYROLITE
C CUKA1.5418  3.21        3.02          0409-449,GYROLITE
C CUKA1.5418  3.12                      1009-449,GYROLITE
C CUKA1.5418  2.80        2.61          0409-449,GYROLITE
C CUKA1.5418  2.80                      0609-449,GYROLITE
C CUKA1.5418  2.42                      0209-449,GYROLITE
C CUKA1.5418  2.31        2.03          0209-449,GYROLITE
C CUKA1.5418  2.17                      0209-449,GYROLITE
C CUKA1.5418  2.06                      0209-449,GYROLITE
C CUKA1.5418  1.90                      0209-449,GYROLITE
C CUKA1.5418  1.82        1.73          0109-449,GYROLITE
C CUKA1.5418  1.57                      0309-449,GYROLITE
C CUKA1.5418  1.52        1.50          0109-449,GYROLITE
C CUKA1.5418  1.50                      0209-449,GYROLITE
C CUKA1.5418  1.41        1.37          0109-449,GYROLITE
C CUKA1.5418  1.37                      0209-449,GYROLITE
C CUKA1.5418  1.34                      0309-449,GYROLITE
C CUKA1.5418  1.28                      0109-449,GYROLITE
C CUKA1.5418  1.195                     0309-449,GYROLITE
C CUKA1.5418  1.168       1.154         0209-449,GYROLITE
C CUKA1.5418  1.060                     0209-449,GYROLITE
C CUKA1.5418  1.040                     0209-449,GYROLITE
C CUKA1.5418  0.96                      0059-449,GYROLITE
C CUKA1.5418  0.93                      0059-449,GYROLITE
C CUKA1.5418  0.908                     0209-449,GYROLITE
C CUKA1.5418  0.882                     0209-449,GYROLITE
C END
C
C   PLOTTING IS DONE VIA CALLS TO THE STANDARD CALCOMP ROUTINES:
C   PLOTS,PLOT,NUMBER AND SYMBOL.
C
```

```
C...LR IS THE STREAM NUMBER FOR INPUTTING DATA
C...NP IS THE NUMBER OF MINERALS OR MIXTURES TO BE PLOTTED.
C...SC IS READ AS 'SCAL' IF SCALES ARE REQUIRED TO BE PLOTTED.
C...AMPSCL IS THE TYPE OF AMPLITUDE SCALE FOR PLOTTING INTENITIES
C   IT IS READ AS 'SQRT', 'LOG' OR BLANK (FOR LINEAR)
C...DIAM IS THE CAMERA DIAMETER IN CMS.
C...TITLE IS READ FROM THE DATA CARDS AND PLOTTED ABOVE THE GRAPH.
C...RAD IS A FOUR BYTE STRING INDICATING THE TYPE OF RADIATION USED.
C...WAVE IS THE WAVELENGTH IN ANGSTROMS OF THE RADIATION USED.
C...ZD IS THE LATTICE SPACING IN ANGSTROMS OF A 'LINE'.
C...ZDA IS THE SECOND LATTICE SPACING REQUIRED TO DEFINE
C   THE LIMITS OF A BROAD BAND.
C...ZII IS THE INTENSITY OF A LINE NORMALISED TO 100.
C...X IS THE X-COORDINATE IN CMS. OF A LINE OR TICK MARK.
C...XA IS THE X CO-ORDINATE IN CMS. OF A LINE WHICH IS PART OF
C   A BROAD BAND
C...Y IS THE HEIGHT IN CMS. OF A TICK REPRESENTING A LINE.
C...BUFFX AND BUFFY SAVE THE X AND Y VALUES OF TICK MARKS
C   REPRESENTING LINES TO BE PLOTTED TO THE  LEFT OF THE CENTRE LINE.
C...BUFFB KEEPS TAGS ON WHICH LINES ARE OF TYPE ZDA.
C...B IS THE VERTICAL LEVEL OF THE SCALE AXIS RELATIVE TO THE ORIGIN.
C...M KEEPS A TALLY OF THE NUMBER OF LINES TO BE PLOTTED TO THE
C   LEFT OF THE CENTRE LINE.
      DIMENSION TITLE(13),BUFFX(100),BUFFY(100)
      REAL LOG,LINEAR(2),SCLTIT(4),SCLTYP(2),SQURT(2)
      LOGICAL BUFFB(100)
      DATA END/'END '/,GAP/'    '/,YES/'SCAL'/,LOG/'LOG '/,ROOT/'SQRT'/
      DATA SQURT/'SQ. ','ROOT'/,LINEAR/'LINE','AR  '/
      DATA SCLTIT/'INTE','NSIT','Y SC','ALE:'/
      NF=0
      LR=4
      DIAM=11.483
C  INITIALISE PLOTTING AND SET ORIGIN
      CALL PLOTS(8.0,2.0,0)
C  READ CONTROL PARAMETERS
      READ(LR,101) NP,SC,AMPSCL
  101 FORMAT(I2,1X,A4,2X,A4)
C  SET INTENSITY SCALE ANNOTATION
      NCHAR=8
      IF (AMPSCL.NE.LOG) GO TO 2
      SCLTYP(1)=LOG
      NCHAR=4
      GO TO 4
    2 IF (AMPSCL.NE.ROOT) GO TO 3
      SCLTYP(1)=SQURT(1)
      SCLTYP(2)=SQURT(2)
      GO TO 4
    3 SCLTYP(1)=LINEAR(1)
      SCLTYP(2)=LINEAR(2)
      GO TO 4
C
C  READ DATA AND PLOT NP GRAPHS
C
    4 DO 70 I=1,NP
      M=0
      B=FLOAT(I-1)*5.0
      DO 5 J=1,100
      BUFFB(J)=.FALSE.
    5 CONTINUE
C  PLOT CENTRE LINE OF LENGTH CNTHT
      CNTHT=3.0
      IF (SC.NE.YES) CNTHT=2.0
      CALL PLOT(0.0,B,3)
      CALL PLOT(0.0,(B+CNTHT),2)
   10 READ(LR,100)RAD,WAVE,ZD,ZDA,ZII,(TITLE(J),J=1,11)
  100 FORMAT(A4,F8.0,2F10.0,F4.0,11A4)
      IF (NF.EQ.1) GO TO 30
C  NF IS A SWITCH TO DETERMINE WHEN THE SCALE SHOULD BE PLOTTED
C  IF IT IS REQUIRED
      IF (SC.NE.YES) GO TO 20
      CALL XRSCAL(B,WAVE,DIAM)
   20 TITLE(12)=GAP
      TITLE(13)=RAD
      CALL SYMBOL(0.0,(3.5+B),0.4,TITLE,0.0,52)
      NF=1
   30 IF (RAD.EQ.END) GO TO 50
C  CHECK THAT LINE REPRESENTS A POSSIBLE REFLECTION
      IF (WAVE.GT.(2.*ZD) .OR. (ZDA.GT.0..AND.WAVE.GT.(2.*ZDA)))
     +      GO TO 10
C
C  CALCULATE POSITION AT WHICH LINE IS TO BE PLOTTED
      X=ASIN(WAVE/(2.*ZD))*DIAM
C
C  IF A BROAD BAND IS TO BE PLOTTED, FIND POSITION
C  OF SECOND LIMIT
      IF (ZDA.GT.0.0001) XA=DIAM*ASIN(WAVE/(2.*ZDA))
C  CALCULATE TICK HEIGHT
      IF (AMPSCL.NE.LOG) GO TO 33
      Y=ALOG10(ZII)*0.5
      GO TO 38
   33 IF (AMPSCL.NE.ROOT) GO TO 35
      Y=SQRT(ZII)*0.1
```

```
      GO TO 38
   35 Y=ZII/100.
C  CHECK WHETHER LINE IS TO BE SAVED FOR PLOTTING TO LEFT OF CENTRE
   38 IF (ZD.LT.2.5) GO TO 40
      M=M+1
      BUFFX(M)=X*(-1.)
      BUFFY(M)=Y
      IF (ZDA.LT.0.0001) GO TO 40
      BUFFB(M)=.TRUE.
      M=M+1
      BUFFX(M)=XA*(-1.)
      BUFFY(M)=Y
   40 CONTINUE
C  PLOT TICK
      CALL PLOT(X,(1.5+B),3)
      CALL PLOT(X,Y+(1.5+B),2)
C  CHECK IF BROAD BAND TO BE PLOTTED
      IF (ZDA.LT.0.00001) GO TO 10
      CALL PLOT(XA,Y+(1.5+B),2)
      CALL PLOT(XA,(1.5+B),2)
      GO TO 10
   50 CONTINUE
C  PLOT TICKS TO LEFT
      DO 60 J=1,M
      CALL PLOT(BUFFX(J),(1.5+B),3)
      CALL PLOT(BUFFX(J),BUFFY(J)+(1.5+B),2)
      IF (BUFFB(J).NE..TRUE.) GO TO 60
      CALL PLOT(BUFFX(J+1),BUFFY(J)+(1.5+B),2)
   60 CONTINUE
      NF=0
C  PLOT INTENSITY SCALE TYPE
      CALL SYMBOL(-3.5,B+1.0,0.2,SCLTIT,0.0,16)
      CALL SYMBOL(-3.0,B+0.65,0.2,SCLTYP,0.0,NCHAR)
   70 CONTINUE
C  TERMINATE PLOTTING
      CALL PLOT(0.0,0.0,999)
      STOP
      END
      SUBROUTINE XRSCAL(B,WAVE,DIAM)
C...PLOTS AN X-RAY DIFFRACTION FILM SCALE, ANNOTATED IN ANGSTROM
C   UNITS, FOR A PARTICULAR WAVELENGTH (WAVE),AND WITH ITS
C   AXIS AT LEVEL B WITH RESPECT TO THE ORIGIN.
C...D IS THE HEIGHT IN CMS. OF A LARGE TICK.
C...S IS THE HEIGHT IN CMS. OF A SMALL TICK.
C...SIZE IS THE HEIGHT (CMS.) OF THE ANNOTATION, AND IS
C   PROPORTIONAL TO THE WAVELENGTH, AND THEREFORE TO THE SIZE
C   OF THE INDIVIDUAL PLOTS.
      D=0.7
      S=0.35
      SIZE=0.2*WAVE/1.5418
C  PLOT TICKS FOR SCALE VALUES 25, 20 AND 15
      DO 5 J=25,15,-5
      X=DIAM*ASIN(WAVE/(2.*FLOAT(J)))
      CALL PLOT(X,B,3)
      CALL PLOT(X,(B+D),2)
      IF (J.NE.15) GO TO 5
      CALL NUMBER((X-0.2),(B-0.3),SIZE,FLOAT(J),0.,-1)
    5 CONTINUE
C  PLOT MORE TICKS ON SCALE
      CALL TICK5(10.,1.,B,D,S,WAVE,SIZE,DIAM)
      CALL TICK2(5.,0.5,B,D,S,WAVE,SIZE,DIAM)
      CALL TICK2(4.,0.5,B,D,S,WAVE,SIZE,DIAM)
      CALL TICK5(3.,0.1,B,D,S,WAVE,SIZE,DIAM)
      CALL TICK5(2.5,0.1,B,D,S,WAVE,SIZE,DIAM)
      CALL TICK5(2.,0.1,B,D,S,WAVE,SIZE,DIAM)
      CALL TICK5(1.5,0.1,B,D,S,WAVE,SIZE,DIAM)
      CALL TICK2(1.,0.05,B,D,S,WAVE,SIZE,DIAM)
      CALL TICK2(0.9,0.05,B,D,S,WAVE,SIZE,DIAM)
      CALL TICK2(0.8,0.01,B,D,S,WAVE,SIZE,DIAM)
      ARGU=WAVE/(2.*0.78)
      IF (ARGU.GE.1.) GO TO 10
      X=DIAM*ASIN(ARGU)
      CALL PLOT(X,B,3)
      CALL PLOT(X,(B+S),2)
      CALL NUMBER((X-0.25),(B-0.3),SIZE,0.78,0.,2)
   10 X=DIAM*3.14159*0.5
      CALL PLOT(X,B,3)
      CALL PLOT(0.0,B,2)
      RETURN
      END
      SUBROUTINE TICK2(FT,SINT,B,D,S,WAVE,SIZE,DIAM)
C...PLOTS AND ANNOTATES A LARGE TICK AT SCALE VALUE FT AND
C   PLOTS A SMALL TICK AT (FT+SINT).
C...BS IS A VALUE USED IN POSITIONING TEXT AGAINST TICKS
C...ND INDICATES NUMBER OF DECIMAL PLACES IN TEXT (-1=NONE)
      ARGU=WAVE/(2.*FT)
      IF (ARGU.GE.1.) RETURN
      X=DIAM*ASIN(ARGU)
      CALL PLOT(X,B,3)
      CALL PLOT(X,(B+D),2)
      BS=0.1
      ND=-1
```

```
      IF ((FT.EQ.1.5).OR.(FT.LT.1.)) BS=0.25
      IF ((FT.EQ.1.5).OR.(FT.LT.1.)) ND=1
      CALL NUMBER((X-BS),(B-0.3),SIZE,FT,0.,ND)
      FTT=FT-SINT
      ARGU=WAVE/(2.*FTT)
      IF (ARGU.GE.1.) RETURN
      X=DIAM*ASIN(ARGU)
      CALL PLOT(X,B,3)
      CALL PLOT(X,(B+S),2)
      RETURN
      END
      SUBROUTINE TICK5(FT,SINT,B,D,S,WAVE,SIZE,DIAM)
C...PLOTS AND ANNOTATES A LARGE TICK AT SCALE VALUE FT AND
C   PLOTS FOUR SMALL TICKS AT SUBSEQUENT INTERVALS OF SINT.
      ND=-1
      BS=0.1
      ARGU=WAVE/(2.*FT)
      IF (ARGU.GE.1.) RETURN
      X=DIAM*ASIN(ARGU)
      CALL PLOT(X,B,3)
      CALL PLOT(X,(B+D),2)
      IF ((FT.EQ.1.5).OR.(FT.EQ.2.5)) ND=1
      IF((FT.EQ.1.5).OR.(FT.EQ.2.5)) BS=0.2
      CALL NUMBER((X-BS),(B-0.3),SIZE,FT,0.,ND)
      DO 20 J=1,4
      FTT=FT-(FLOAT(J)*SINT)
      ARGU=WAVE/(2.*FTT)
      IF (ARGU.GE.1.) RETURN
      X=DIAM*ASIN(ARGU)
      CALL PLOT(X,B,3)
   20 CALL PLOT(X,(B+S),2)
      RETURN
      END
```